Integral Convergence Test for Infinite Series

Let's explore the integral convergence test and its role in approximating sums of infinite series.

```
> restart;
```

Maple will sum some infinite series for us. Take for example the harmonic and alternating harmonic series.

```
> sum(1/k,k=1..infinity);
  sum((-1)^k/k,k=1..infinity);
```

$$\infty$$
$$-\ln(2) \tag{1}$$

The harmonic series diverges whereas the alternating harmonic series converges (to -ln(2)).
When it is impossible to get an exact answer we can approximate series sums by choosing a large upper limit. For example, summing the first 10,000 terms of the alternating harmonic series doesn't take much time at all and this gives a fairly accurate answer (the exact answer is -ln(2)).

```
> evalf(sum((-1)^k/k,k=1..10000));
  evalf(-ln(2));
```

$$-0.6930971831$$
$$-0.6931471806 \tag{2}$$

Of course, we would like to know how far off an estimate might be. We can get a handle on the accuracy of an estimate (for a series whose terms are given by a positive decreasing function) by using the integral test.

Recall that when f(x) is a positive decreasing function with f(k) = a_k and sum(a_k) is a convergent series then...

```
> Int(f(x),x=N..infinity) <= Sum(a[k],k=N..infinity) and
  Sum(a[k],k=N..infinity) <= a[N] + Int(f(x),x=N..infinity);
```

$$\int_N^\infty f(x)\,\mathrm{d}x \le \sum_{k=N}^{\infty} a_k \le a_N + \int_N^\infty f(x)\,\mathrm{d}x \tag{3}$$

For a convergent series (where we are allowed to apply the integral test) this gives an error estimate for our tail.

Consider the series...

```
> sum(arctan(k)/k^2,k=1..infinity);
```

$$\sum_{k=1}^{\infty} \frac{\arctan(k)}{k^2} \tag{4}$$

We can see (looking at the following plot) that the terms of this series are given by a positive decreasing function, so we can apply the integral test.

```
> f := x -> arctan(x)/x^2:
  'f(x)' = f(x);
```

(5)

$$f(x) = \frac{\arctan(x)}{x^2} \tag{5}$$

```
> plot(f(x),x=1..10);
```



Looking at the following integral, we see that our series converges.

```
> Int(f(x),x=1..infinity) = int(f(x),x=1..infinity);
```

$$\int_1^{\infty} \frac{\arctan(x)}{x^2}\, dx = \frac{\pi}{4} + \frac{\ln(2)}{2} \tag{6}$$

In fact, the integral test gives us the following bounds on the series sum:

```
> lowerBound := int(f(x),x=1..infinity);
  upperBound := arctan(1)/1^2 + int(f(x),x=1..infinity);
```

$$lowerBound := \frac{\pi}{4} + \frac{\ln(2)}{2}$$

$$upperBound := \frac{\pi}{2} + \frac{\ln(2)}{2} \tag{7}$$

```
> evalf(lowerBound);
  evalf(upperBound);
```

1.131971754

1.917369917 **(8)**

Let's find N so that sum(arctan(k)/k^2,k=1..N) is accurate with error less than 0.001.
In other words, we need to find N so that...

```
> sum(arctan(k)/k^2,k=N+1..infinity) <= 0.001;
```

$$\sum_{k=N+1}^{\infty} \frac{\arctan(k)}{k^2} \leq 0.001 \quad (9)$$

We know that this remainder is bounded by an indefinite integral. So we should try to find N so that...

```
> int(f(x),x=N..infinity) <= 0.001;
```

$$\frac{\sqrt{\frac{1}{N^2}}\left(N^2 \ln\left(\frac{N^2+1}{N^2}\right)\sqrt{\frac{1}{N^2}} + \pi N \sqrt{\frac{1}{N^2}} - 2 \arctan\left(\sqrt{\frac{1}{N^2}}\right)\right)}{2} \leq 0.001 \quad (10)$$

Let's solve for N.

Note: I will use "fsolve" to find an approximate solution. Trying to get Maple to "solve" and find an exact solution, makes Maple choke.

```
> Eqn := int(f(x),x=N..infinity) = 0.001;
```

$$Eqn := \frac{\sqrt{\frac{1}{N^2}}\left(N^2 \ln\left(\frac{N^2+1}{N^2}\right)\sqrt{\frac{1}{N^2}} + \pi N \sqrt{\frac{1}{N^2}} - 2 \arctan\left(\sqrt{\frac{1}{N^2}}\right)\right)}{2} = 0.001 \quad (11)$$

```
> fsolve(Eqn, N);
```

$$1570.477952 \quad (12)$$

It looks like N = 1571 should give us the accuracy we want.

```
> evalf(sum(arctan(k)/k^2,k=1..1571));
```

$$1.606282604 \quad (13)$$

So our series sum is somewhere between 1.606282605 and 1.606282605 + 0.001 = 1.607282605

Let's see what the sum of the first 100,000 terms gives us...

```
> evalf(sum(arctan(k)/k^2,k=1..100000));
```

$$1.607260041 \quad (14)$$

The bound on the error of this approximation is...

```
> evalf(int(f(x),x=100000..infinity));
```

$$0.00001570486327 \quad (15)$$